Micro-Professor Application Note

—DOC. NO. MPF-I-02-210A —

MPF-I AS A TRAFFIC LIGHT CONTROLLER

An Application Example of Z8Ø-PIO.

MULTITECH INDUSTRIAL CORPORATION

OFFICE : 977, MIN SHEN E. ROAD, TAIPEI, 105, TAIWAN, R.O.C.
TEL: (02)769-1225 (10 LINES) TELEX: 23756 MULTIIC

FACTORY: 5, TECHNOLOGY ROAD III
HSINCHU SCIENCE-BASED INDUSTRIAL PARK.
HSINCHU, TAIWAN, 300, R.O.C.
TEL: (035)775102 (3 LINES)

US$1.00

Purpose: Use PIO for traffic light control

Required Equipment: A PIO chip, a 75492, three LED lamps (one in green, one in red, and one in yellow), three resistors, and some wire.

You are required to use the necessary devices to make the hardware connections in accordance with the diagram shown below:



fig A

Expriment Explanation:

1. The PIO is a 40-pin large-scale integrated circuit (LSI) especially designed to provide TTL compatible interface between peripheral devices and the Z80 CPU. The CPU can configure the Z80-PIO to interface with a wide range of peripheral devices with no other external logic required. Typical peripheral devices that are fully compatible with the Z80-CPU include most keyboard, paper tape readers and punches, printers, and PROM programmers, etc. It is programmable. The PIO has two I/O ports--port A and port B. Each port is connected to eight pins. The addresses of the PIO are from 80 to 83 (in hexadecimal). In this experiment, port A will be used. For detailed description of the PIO and its operation, refer to "Z80 Microprocessor Programming and Interfacing, Book 2" by Nichols, Rony, published by Blacksburg; or Z80 Handbook.

2. Each of the two ports of the PIO has four modes of operation; namely, byte output, byte input, byte bidirectional bus, and bit control mode. The mode of operation must be established by writing a control word to the PIO in the following format:

D7	D6	mode of operation
ϕ	ϕ	Byte output
ϕ	1	Byte input
1	ϕ	Byte bidirectional
1	1	Bit control

fig B

We can change the contents of bit D7 and D6 to form a control word in order to change the mode of operation of port A.

3. In this experiment, the mode of operation of port A is byte output. Thus, the contents of bit D7 and D6 should be zero, and the contents of bit D3 through bit D0 should be one. The contents of bit D5 and D4 make no difference to the control word.

D7	D6	D5	D4	D3	D2	D1	Dϕ
M1	Mϕ	×	×	1	1	1	1

control word

4. Of the four addresses of PIO, two addresses are assigned to port A--80H is used as the data port of port A, and 82H is used as the control port of port A. Since we use port A in its byte output mode, the control word is set 00001111(binary) (or 0FH). The value of the control word should be sent to the control port of Port A to set Port A to its byte output mode.

5. We use the bit Ø (AØ) of Port A to control the green light, A1 to control the yellow light, and A2 to control the red light. To illuminate the red light, the value Ø1 should be sent to the data port of PIO (whose address is 8ØH). By sending ØlH to the data port of PIO, the eight bits on the Port A will become

A7 A6 A5 A4 A3 A2 A1 AØ

Ø Ø Ø Ø Ø Ø Ø 1

The 75492 will convert the input from AØ to low, so the output at pin 1Y of 75492 is low. This will cause the electrical current to flow from the resistor to the green LED lamp.

To illuminate the yellow LED, the byte (Ø2H) should be sent to the data port of the PIO. This byte will cause the A1 high and 2Y low. To illuminate the red lamp, the byte (Ø4H) is sent to the data port of the PIO.

6. For how long will a lamp be illuminated? This is controlled by time delay subroutines--DELAY, DELAY1, and DELAY2.

Since the MPF-I operates at 1.79MHz, a T state is Ø.56 micro-seconds. Therefore, the time delay achieved by the DELAY subroutine is

Ø.56 micro-seconds x {7+4[10+(16+4+4+10)x65536+4+12]-5+10}=4.9912867 sec

And the time delay for DELAY1 is

Ø.56 x [10+(16+4+4+11+12)x19000]=Ø.5000856 sec

The time delay for DELAY2 is

Ø.56 x [10+(16+4+4+11+12)x65536]=1.7249131 sec



```
LOC    OBJ CODE M SOURCE STATEMENT

1800               ORG     1800H
           PIODA   EQU     80H          ;DATA PORT OF PIO
                                         CHANNEL A
           PIOCA   EQU     82H          ;CONTROL PORT OF PIO
                                         CHANNEL A
           START:
1800   3E0F        LD      A,0FH
1802   D382        OUT     (PIOCA),A    ;PIO PORT A OUTPUT MODE
           BEGIN:
1804   3E01        LD      A,01H
1806   D380        OUT     (PIODA),A    ;GREEN LED LIGHT
1808   CD3018      CALL    DELAY        ;DELAY 5 SEC
180B   0604        LD      B,4
           FLASH:
180D   C5          PUSH    BC
180E   3E00        LD      A,0
1810   D380        OUT     (PIODA),A    ;FLASH 4 SEC
1812   CD4018      CALL    DELAY1
1815   3E01        LD      A,01
1817   D380        OUT     (PIODA),A
1819   CD4018      CALL    DELAY1
181C   C1          POP     BC
181D   10EE        DJNZ    FLASH
           ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
181F   3E02        LD      A,02
1821   D380        OUT     (PIODA),A    ;YELLOW LED LIGHT
1823   CD4A18      CALL    DELAY2       ;1.725 SEC
1826   3E04        LD      A,04
1828   D380        OUT     (PIODA),A    ;RED LED LIGHT
182A   CD3018      CALL    DELAY        ;5 SEC
182D   C30418      JP      BEGIN
           ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
           ;
           ;       DELAY 4 * 1.25 SEC = 5 SEC  SUBROUTINE
           ;       0.56 us * (16 + 4 + 4 + 10 ) * 65536 =1.25 SEC
           ;
           ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
           DELAY:
1830   1604        LD      D,4          ; 7T
           DELX:
1832   010000      LD      BC,0         ; 10T
           DE0:
1835   EDA1        CPI                  ; 16T
1837   00          NOP                  ; 4T
1838   00          NOP                  ; 4T
1839   EA3518      JP      PE,DE0       ; 10T
183C   15          DEC     D
183D   20F3        JR      NZ,DELX
183F   C9          RET
           ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
           ;
           ;       DELAY 0.5 SEC SUBROUTINE
           ;       0.56 us * (16 + 4 + 4 + 11 + 12 ) * 19000 =0.5 SEC
           ;
           ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
           DELAY1:
1840   01384A      LD      BC,4A38H
           DE1:
1843   EDA1        CPI
```



```
LOC    OBJ CODE M SOURCE STATEMENT

1845   00          NOP
1846   00          NOP
1847   E0          RET     PO
1848   18F9        JR      DE1
           ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
           ;
           ;       DELAY 1.725 SEC SUBROUTINE
           ;
           ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
           DELAY2:
184A   010000      LD      BC,0
           DE2:
184D   EDA1        CPI
184F   00          NOP
1850   00          NOP
1851   E0          RET     PO
1852   18F9        JR      DE2
```